
Appendix B

SPNE derivation

assumptions

In[1]:= A0 = And[t > 0, g ≥ 0, r > 0, c1 ≥ 0, c2 ≥ 0, a1 ≥ 0, a2 ≥ 0, n ≥ 1, v1 ≥ 0, v2 ≥ 0];
A0b = And[t > 0, g ≥ 0, r > 0, c1 ≥ 0, c2 ≥ 0, a1 ≥ 0, a2 ≥ 0, n > 0, v1 ≥ 0, v2 ≥ 0];
A1 = And[v1 + v2 > c1 + c2 + 6 t, v1 > c1];
A2 = (-r (g + r) / (2 n) + 2 t ≥ c2 - c1 + v1 - v2 ≥ r (-2 g + (-1 + 2 n) r) / (4 n) - 6 t);

A3 = (1 + 2 n) r2 - 32 n t < 0;

A4 = (v1 - c1 ⩵ v2 - c2);
A5 = (g < (-7 + 2 n) r / 8);

A6 = v1 - v2 > -t 96 g2 + 16 g (11 - 2 n) r + (-7 + 2 n) (-11 + 2 n) r2 - 16 n t 

((1 + 2 n) r (-16 g + (-7 + 2 n) r) + 32 n (-7 + 2 n) t);

define utility functions (skip the retailer index j for convenience)

In[9]:= V1 = v1 + r * a1 + g * (a1 + a2); V2 = v2 + r * a2 + g * (a1 + a2);
u1 = V1 - p1 - t * x;
u2 = V2 - p2 - t * (1 - x);

indifferent consumer locations (market covered / uncovered)

In[11]:= xCov = Solve[u1 ⩵ u2, x][[1, 1, 2]];
xUn1 = Solve[u1 ⩵ 0, x][[1, 1, 2]]; xUn2 = Solve[u2 ⩵ 0, x][[1, 1, 2]];

profit-maximizing retail prices when market not covered

In[13]:= ProfRUnc = (1 / n) ((p1 - w1) * xUn1 + (p2 - w2) (1 - xUn2)) - a1^2;
NEpUnc = FullSimplify[

Join[Solve[D[ProfRUnc, p1] ⩵ 0, p1][[1]], Solve[D[ProfRUnc, p2] ⩵ 0, p2][[1]]]]

Out[14]= p1 →
1

2
(a2 g + a1 (g + r) + v1 + w1), p2 →

1

2
(a1 g + a2 (g + r) + v2 + w2)



profit-maximizing retail prices when market covered

In[15]:= ProfRCov = (1 / n) ((p1 - w1) * xCov + (p2 - w2) (1 - xCov)) - a1^2;
MaxP2 = Solve[(u1 /. x -> xCov) ⩵ 0, p2][[1]];
ProfRCovOneVar = ProfRCov /. MaxP2;
NEpCov = Solve[D[ProfRCovOneVar, p1] ⩵ 0, p1][[1]];
NEpCov = FullSimplify[Join[NEpCov, MaxP2 /. NEpCov]]

Out[18]= p1 →
1

4
(4 (a1 + a2) g + (3 a1 + a2) r - 2 t + 3 v1 + v2 + w1 - w2),

p2 →
1

4
(4 (a1 + a2) g + (a1 + 3 a2) r - 2 t + v1 + 3 v2 - w1 + w2)

reduced game equilibrium wholesale prices when market not covered

In[19]:= ProfM1Unc = (1 / n) ((w1 - c1) * xUn1) /. NEpUnc;
ProfM2Unc = (((w2 - c2) * (1 - xUn2)) /. NEpUnc) - a2^2;
NEwUnc = Solve[w1 ⩵ c1 && D[ProfM2Unc, w2] ⩵ 0, {w1, w2}][[1]]

Out[20]= w1 → c1, w2 →
1

2
(c2 + a1 g + a2 g + a2 r + v2)

reduced game equilibrium wholesale prices when market covered

In[21]:= ProfM1Cov = (1 / n) ((w1 - c1) * xCov);
ProfM2Cov = (((w2 - c2) * (1 - xCov))) - a2^2;
NEwCov = Solve[w1 ⩵ c1 && D[(ProfM2Cov /. NEpCov), w2] ⩵ 0, {w1, w2}][[1]]

Out[22]= w1 → c1, w2 →
1

2
(c1 + c2 - a1 r + a2 r + 2 t - v1 + v2)

uncovered market possible?

In[23]:= Reduce[And[Or[((xUn1 < xUn2) /. NEpUnc /. NEwUnc),
((xUn1 < xUn2) /. NEpUnc /. NEwCov)], A0, A1]]

Out[23]= False

reduced game equilibrium advertising when market covered

profit functions concave?

In[24]:= ProfRStage1 = (ProfRCov /. NEpCov /. NEwCov);
ProfM2Stage1 = (ProfM2Cov /. NEpCov /. NEwCov);
Simplify[And[D[ProfRStage1, {a1, 2}] < 0], And[A0, A1, A2, A3]]

Out[25]= True

equilibrium advertising

In[26]:= NEa =

FullSimplify[Solve[D[ProfRStage1, a1] ⩵ D[ProfM2Stage1, a2] ⩵ 0, {a1, a2}][[1]]]

Out[26]= a1 →
c1 r - c2 r + g r2 + r3 - 16 g t - 14 r t - r v1 + r v2

r2 + 2 n r2 - 32 n t
,

a2 →
r (r (g + r) - 2 n (c1 - c2 + 2 t - v1 + v2))

(1 + 2 n) r2 - 32 n t


corner solution (or negative advertising) possible?

In[27]:= xNE = FullSimplify[(xCov /. NEpCov /. NEwCov /. NEa)];
Reduce[And[Not[0 ≤ xNE ≤ 1], A0, A1, A2, A3], Reals]
Reduce[And[Or[NEa[[1, 2]] < 0, NEa[[2, 2]] < 0], And[A0b, A1, A2, A3]]]

Out[28]= False

Out[29]= False

Comparative Statics

Proposition 1

In[30]:= pDne = Simplify[((p1 - p2) /. NEpCov /. NEwCov /. NEa)];
wDne = Simplify[((w1 - w2) /. NEwCov /. NEa)];
aDne = Simplify[((a1 - a2) /. NEa)];

In[33]:= Simplify[And[D[aDne, c1] < 0, D[xNE, c1] < 0], And[A0, A1, A2, A3]]

Out[33]= True

In[34]:= Simplify[And[D[aDne, c2] > 0, D[xNE, c2] > 0], And[A0, A1, A2, A3]]

Out[34]= True

In[35]:= Simplify[And[D[pDne, g] > 0, D[wDne, g] > 0, D[aDne, g] > 0, D[xNE, g] > 0],
And[A0, A1, A2, A3]]

Out[35]= True

In[36]:= Reduce[Not[And[D[pDne, n] < 0, D[wDne, n] < 0, D[aDne, n] < 0, D[xNE, n] < 0]] &&
And[A0b, A1, A2, A3]]

Out[36]= False

In[37]:= Simplify[And[D[pDne, v1] > 0, D[wDne, v1] > 0, D[aDne, v1] > 0, D[xNE, v1] > 0],
And[A0, A1, A2, A3]]

Out[37]= True

In[38]:= Simplify[And[D[pDne, v2] < 0, D[wDne, v2] < 0, D[aDne, v2] < 0, D[xNE, v2] < 0],
And[A0, A1, A2, A3]]

Out[38]= True

Proposition 2 (prove negation is false)

In[39]:= Reduce[Not[And[D[pDne, r] < 0, D[wDne, r] < 0, D[aDne, r] < 0, D[xNE, r] < 0]] &&
And[A0, A1, A2, A3, A4, A5]]

Out[39]= False

Under A4 xNE is independent of innate valuations and costs

In[40]:= FullSimplify[xNE, A4]

Out[40]= -
r (2 g + r - 2 n r) + 24 n t

(1 + 2 n) r2 - 32 n t

Proposition 3 (prove negation is false)

In[41]:= R = FullSimplify[pDne / aDne, A4];
Reduce[And[D[R, r] < 0, A0b, A1, A2, A3, A4, A5, A6], Reals]

Out[42]= False

